==============LC_Simple_Q========================



Simple_Q

```
*  http://www.idea2ic.com/PlayWithJavascript/L_C_R_F.html
*                                   ___                                  ___
*  XL = XC = 1 Ohms @1Hz __________|LP |                      __________|BP |
*                       |          |___|                     |          |___|
*             _  _  _   |                                    |  _  _  _
*       VIN  / \/ \/ \  |      R1        VIN          R2     | / \/ \/ \
*        ____| () () |__|_/\  /\  /\_         ___/\  /\  /\__|_| () () |_
*      _|___           |    \/  \/  _|_     _|___  \/  \/   |      L2   _|_
*     /_   \    L1    _|_           ///    /_   \          _|_          ///
*    // \   \         ___ C2              // \   \         ___ C2
*    \   \_//          |                  \   \_//          |
*     \____/          _|_                  \____/          _|_
*      _|_            ///                   _|_            ///
*      ///                                  ///
*          H(s) = 1/(s^2 +s*w0/Q + w0^2)          dsauersanjose@aol.com  4/15/08
```

```
.OPTIONS  GMIN=1e-15    METHOD=gear   ABSTOL=1e-15  TEMP=27 srcsteps = 1  gminsteps = 1
VIN       VIN     0     DC    0V      PWL(0 1 2 1 2.1 0)                  AC     1V
L1        VIN     LP    .159
C1        LP      0     .159
R1        LP      0     10
R2        VIN     BP    10
L2        BP      0     .159
C2        BP      0     .159
```

```
*=======Q_Or_Quality_Factor_Tells_A_Lot=========
.control

*#1=======Q_is_the_Number_Of_Cycles_To_Settle_Out=========
tran      1       20      0
run
plot      lp      title   Settle_Time_Q_10

*#2=======Q_Also_Defines_A_Lowpass_Filters_Peaking=========
ac        dec     200     .1      10
run
plot      db(lp)  title   LowPassPeaking_Q_10

*#3=======Q_Also_Defines_the_Bandwidth_Of_A_BandPass=========
plot      db(bp)  linear  ylimit -3  0   xlimit  .9  1.1  title  BandPassBW_Q_10
```

```
*#4=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
alter      R1     resistance    =   1
alter      R2     resistance    =   1

tran       1      20      0
run
plot       lp       title   Settle_Time_Q_1
*#6=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
ac         dec      200      .1      10
run
plot       db(lp)   title   LowPassPeaking_Q_1
*#7=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
plot       db(bp)   linear  ylimit -3  0   xlimit  .1  2 title  BandPassBW_Q_1
.endc
.end


======================END_OF_SPICE============================
#1=======Q_is_the_Number_Of_Cycles_To_Settle_Out=========
tran       1      20      0
run
plot       lp       title   Settle_Time_Q_10
```

The settle time for Q=10 is about 10 cycles




```
*#2=======Q_Also_Defines_A_Lowpass_Filters_Peaking=========
ac         dec      200      .1      10
run
plot       db(lp)   title   LowPassPeaking_Q_10
```

The lowpass peaking for Q=10 is about a factor of 10




```
*#3=======Q_Also_Defines_the_Bandwidth_Of_A_BandPass=========
plot       db(bp)     linear  ylimit -3  0   xlimit  .9  1.1  title  BandPassBW_Q_10
```

The bandwidth for Q=10 is 1/10th center freq




```
*#4=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
alter      R1     resistance    =   1
alter      R2     resistance    =   1

tran       1        20      0
run
plot       lp       title   Settle_Time_Q_1
```

The settle time for Q=1 is about 1 cycles




```
*#6=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
ac          dec       200       .1       10
run
plot        db(lp)    title     LowPassPeaking_Q_1
```

The lowpass peaking for Q=1 is about a factor of 1


```
*#7=======Q_is_Reduced_From_10_to_1_to_Show_the_Effects=========
plot        db(bp)    linear    ylimit -3  0   xlimit  .1  2 title  BandPassBW_Q_1
```

The bandwidth for Q=1 is about center freq


Graph 96 - ac306: BandPassBW_Q_1
dB
db(bp)
0.0
-0.5
-1.0
-1.5
-2.0
-2.5
-3.0
decibel
0.0
0.2
0.4
0.6
0.8
1.0
1.2
1.4
1.6
1.8
2.0
frequency Hz